[bookmark: _Toc94001823]


Chapitre e2


Les formules sous

Excel

Dans une cellule active vierge, on peut vouloir calculer une valeur à l’aide d’une formule. Il suffit d’utiliser la syntaxe suivante  dans la cellule active :
=<la formule>
Exemple	= 2*9
La valeur affichée dans la cellule est 18 et dans la barre de saisie  =2*9.
On tape  =  et la formule ; l’écriture étant terminée on appuie sur Entrée.

Attention de ne pas utiliser les flèches qui ont une autre fonction !

Ci-dessous on rappelle les opérateurs arithmétiques simples :

Opérateurs disponibles

	Opérateur
	Opérateur
	Niveau de priorité

	Addition
	+
	1

	Soustraction
	-
	1

	Multiplication
	*
	2

	Division
	/
	2

	Puissance
	^
	3

	Priorité absolue de calcul
	( )
	4

	Concaténation de texte
	&(voir fonction concatener)
	1

	Le reste
d’une division entière
	MOD(nombre;diviseur)
	1




Remarque
Excel calcule la formule en effectuant d’abord de gauche à droite les opérateurs de priorité les plus élevés.
Ainsi =1+2*3 donne 7, alors que =(1+2)*3 donne 9.

La concaténation de texte est un opérateur qui, à partir de 2 chaînes de caractères, en construit une seule contenant l’ensemble des deux. Cet opérateur est classique dans les langages informatiques qui définissent une chaîne de caractères comme un ensemble de lettres et chiffres. Compte tenu qu’il existe dans les langages des mots qui ont une signification particulière, on définit une chaîne en l’encadrant par des guillemets (double cote).

Ainsi :
"bonne" & "journée" donne "bonnejournée"
"bonne " & "journée" donne "bonne journée"
 le dernier caractère de la première chaîne comprend à la fin de « bonne » un espace
qui est pris en compte au même titre que les autres.

Excel utilise cette notation et on peut taper dans une cellule :
="bonne " & "journée"  la cellule affichera bonne journée.

On peut aussi utiliser la fonction CONCATENER :
=CONCATENER("Bonne "; "journée")
Les guillemets peuvent surprendre le lecteur, mais sans les guillemets Excel cherche parmi les noms définis dans la feuille de calcul (comme ceux définissant une sélection).

REFERENCES DES CELLULES

Les références d’une cellule servent à désigner l’emplacement d’une valeur. Il existe deux systèmes dont la compréhension est fondamentale. Ils reposent sur le même principe que la bataille navale, soit une localisation par la colonne et la ligne.

Reprenons l’exemple de l’introduction :

	
	Colonne A
	Colonne B
	Colonne C
	Colonne D
	Colonne E

	Ligne 1
	Année
	2001
	2002
	2003
	2004

	Ligne 2
	Trim 1
	150
	152
	148
	150

	Ligne 3
	Trim 2
	125
	130
	128
	132

	Ligne 4
	Trim 3
	138
	140
	135
	145

	Ligne 5
	Trim 4
	160
	170
	169
	

	Ligne 6
	Total
	573
	592
	580
	



Etant dans la cellule colonne B ligne 6 pour calculer la somme des ventes de l’année 2001 on peut :
utiliser la fonction Somme appliquée aux 4 cellules au-dessus,
ou bien utiliser la fonction Somme appliquée aux cellules de la colonne B, de la ligne 2 à la ligne 5.

On appelle référence relative ou adresse relative l’emplacement d’une cellule mémorisé sous forme de déplacement par rapport à une cellule donnée et référence (ou adresse) absolue l’emplacement d’une cellule par rapport à sa situation dans la feuille.

Remarque
Les deux formulations ci-dessus ne sont pas identiques.
Dans le premier cas les cellules à sommer sont définies relativement à la cellule qu’on veut calculer. On constate que, pour les autres totaux, on fera la somme des 4 cellules au-dessus. La formule sera la même. Il suffira de recopier la même formule, travail facile compte tenu des outils performants de recopie fournis par les tableurs. 
Dans l’autre cas la colonne et les lignes sont fixées. Pour les autres totaux il faudra après recopie modifier le titre de la colonne, modification fastidieuse et source d’erreurs. La localisation des cellules par référence relative intervenant dans une autre cellule donnée est facilement copiable dans une autre cellule. 

Pour l’efficacité du tableur, les références relatives sont évidemment essentielles. Pour l’utilisateur, les références absolues sont les plus claires. Comment faire cohabiter ces deux approches ?
La solution retenue est la suivante : les références sont données en indiquant la colonne et la ligne, le tableur les interprète en références relatives par rapport à la cellule active. Automatiquement le tableur mémorisera l’emplacement en calculant lui-même le déplacement par rapport à la cellule donnée. Pour faire intervenir les références absolues, on ajoute le symbole $ devant les lignes et les colonnes.
L’affichage dans la barre de saisie sera toujours en colonne et ligne (style A1), mais sans $ ; elles sont mémorisées en déplacement relatif. Ainsi l’utilisateur utilise par défaut les références relatives et doit en être conscient contrairement à Monsieur Jourdain.

REFERENCES SOUS EXCEL
On distingue trois types de références correspondant à un comportement différent d’Excel et deux styles d’utilisation. Le repérage par le titre de colonne et le numéro de ligne est dit le style A1, celui qu’on a utilisé jusqu'à présent.

Les références relatives sous Excel ont été utilisées dans le chapitre précédent, en particulier dans le paragraphe 6d) Sélection multiple p.14.
L’exemple ci-dessous illustre l’interprétation des références relatives comme un déplacement. Dans la cellule C2 on a tapé la formule additionnant la valeur de la cellule A1 à 2 :						= A1+2

 (
C2
 contient la formule =
A1
+2 donc la valeur de la cellule 
A1
 +2 soit 14
Mais la mémorisation est de la forme :
«  Prendre la valeur dans la ligne au-dessus, 2 c
olonnes à gauche et ajouter 2 »
) (
A1
 contient 12
A2
 contient 13
)[image: ]

Si on recopie cette formule dans C3, elle sera « prendre la valeur dans la ligne au-dessus 2 colonnes à gauche (donc la cellule A2) et additionner 2 » soit 15. Comme dans un plan d’une ville, on indique dans la formule le lieu d’arrivée et Excel mémorise le chemin à partir de la cellule active, jouant le rôle de lieu de départ. Il travaille selon le principe du GPS.

Si on ne veut pas que Excel mémorise le chemin à partir de la cellule active mais de l’emplacement exact dans la feuille, il faut utiliser les références absolues avec le symbole $ devant chaque titre de colonne et numéro de ligne. Les notations sont alors les suivantes :

pour une cellule
$<titre de la colonne>$<numéro de la ligne>
Exemple	$B$1
pour une ligne
$<numéro de la ligne>:$<numéro de la ligne>
Exemple	$2:$2
pour une colonne 
$<titre de la colonne >:$<titre de la colonne>
Exemple	$A:$A
pour une plage
<référence absolue du coin supérieur gauche>:<référence absolue du coin inférieur droit>
Exemple	$B$1:$C$5

Pour une sélection multiple, les références absolues sont séparées par des points virgules.
Exemple	$A :$A ;$B$2 :$E$1 :$F$3
(sélection multiple en référence absolue comprenant la colonne A la cellule B2 et la plage de E1 à F3)

Note
Les références absolues doivent être utilisées uniquement quand elles sont indispensables, ce qui limite considérablement leur emploi, réduit pratiquement à une seule cellule ou plage.
La touche F4 permet de passer une référence relative en référence absolue ( référence mixte sur la ligne, référence mixte sur la colonne introduites au paragraphe suivant). On tape la référence relative et aussitôt sur F4 jusqu’à obtenir les références souhaitées. 


Exemple
On a en colonne B à partir de la ligne 3 les prix en Euros des biens. La cellule B1 contient la valeur d’un Euro en Francs. On veut calculer les prix en Francs dans la colonne C. On tape la formule en C4 en fixant pour B1 une référence absolue. Ainsi la formule sera recopiée dans les lignes suivantes et donnera le résultat souhaité.

 (
En 
C4
 on a la formule dans la barre de saisie :
=$B$1*B4
On tape  
 
= B 
F4
*flèche gauche (ou cli
c 
sur 
B4
) 
Entrée
La valeur d’un Euro est en référence absolue et 
B4
 en référence relative mémorisé « e
n même ligne colonne à gauche »
Cette formule recopiée en 
C5
,
 
C6
 donne le bon résultat car la valeur 6,55957 est dans 
la cellule en référence absolue
)[image: ]

Elles consistent à fixer la colonne en référence absolue et la ligne en référence relative ou l’inverse. La notation est alors :
$<titre de la colonne><numéro de la ligne>
Exemple	$A2

<titre de la colonne>$<numéro de la ligne>
Exemple	A$2
Exemple utilisant la fonction Somme(<Plage>)

 (
En 
B5
 on tape la formule =somme(
B4
 ;$
G4
)
somme de colonne G fixé ligne au-dessus  à même colon
ne que la cellule de la formule
Cette formule peut être recopiée dans le reste de l
a ligne et même dans la ligne 7
)[image: ]

On peut changer le style A1 en style LC peu employé dit « style ligne colonne » (par bouton Office/ options Excel/Formules et cocher « style de référence L1C1») dans le paragraphe Manipulation de formules. Les lignes et colonnes sont numérotées par des entiers. Avec ces notations une référence absolue est donnée par :

L<numéro>C<numéro>
Exemple	L5C2 équivalent de $B$2

Une référence relative est donnée par les déplacements relatifs codés par des nombres négatifs pour les lignes au-dessus, positifs pour les lignes en-dessous, de même pour les colonnes avec un entier négatif pour une colonne à gauche et positif pour une colonne à droite. Les valeurs du déplacement étant entre parenthèses :

L(déplacement relatif )C(déplacement relatif)
Exemple	=L(-1)C(2)
(correspond à la cellule une ligne au-dessus et deux colonnes à droite)

Les déplacements sont difficiles pour l’utilisateur, qui préfère le style A1. Mais on peut en changeant de style vérifier la pertinence de ce paragraphe.

Sur la cellule active, la saisie d’une formule doit toujours commencer par  = 
=<la formule>
(la cellule affichera la valeur de la formule et la barre de saisie la formule précédée du signe  =)

On entend par formule une expression contenant des opérations, des références à d’autres cellules et l’utilisation de fonctions prédéfinies dans le tableur. Pour taper la formule on peut taper tout au clavier mais on peut, pour les références d’une cellule, sélectionner la cellule désirée (ou la plage de cellules) comme pour une sélection. Les références relatives de la sélection s’écrivent alors directement dans la formule. La validation se fait uniquement avec Entrée (les flèches servant pour les sélections).


Exemple

Pour taper la formule =A1+A2
 je tape =, je clique sur la cellule A1, je tape +,
j’utilise les flèches pour me positionner sur A2, puis Entrée.

Pour taper la formule =somme(F1:F6) (calculant la somme des cellules de la plage F1 à F6)
 je tape =somme(, je sélectionne ma plage, je tape ) puis Entrée.

On peut pratiquer comme pour une donnée mais les modifications des références peuvent se faire plus facilement. La cellule étant active on fait avec la souris un double clic dessus (ou dans la barre de saisie), toutes les cellules apparaissant dans la formule sont bordées avec des couleurs différentes. En s’approchant avec la souris de chaque bordure on obtient une flèche permettant le déplacement par un « cliquer-glisser » ou l’extension de la plage par la poignée apparaissant dans le coin inférieur droit.


 (
En tirant sur cette poignée par un 
« 
cliquer-glisser
 » vers le bas
on étend la sélection
) (
En positionnant la souris sur le pourtour on peut déplacer la plage
)[image: ]

Tout ce qu’on a vu dans le premier chapitre s’applique aux cellules contenant des formules. Lors d’un coller (Ctrl V) les références relatives s’adaptent automatiquement à l’emplacement de la nouvelle cellule, les références absolues n’étant pas modifiées. Le collage spécial permet de recopier uniquement les valeurs et prend donc tout son sens.

On peut aussi en cliquant dans la barre de saisie tout sélectionner et faire une copie (Ctrl C) puis valider ou annuler. En changeant de cellule active on peut faire un coller dans la barre de saisie de l’ancienne cellule qui donnera exactement la même chose sans adapter les références au nouvel emplacement. La validation réinterprétera ces mêmes références.

Sur une ligne ou colonne, on tape la formule dans une cellule qu’on sélectionne et on peut alors utiliser le « cliquer-glisser » (bouton droit ou gauche) comme pour les données. Si les références sont relatives dans la formule, leur affichage sera adapté à la nouvelle cellule. Si elles sont absolues, elles seront copiées sans modification.

On a vu qu’on peut utiliser les fonctions prédéfinies dans les formules. L’utilisateur courant connaît le nom et le mode d’emploi des fonctions qu’il utilise très couramment ou souvent. Il tape alors directement au clavier sa syntaxe. Pour le néophyte, il existe divers moyens lui permettant de trouver les fonctions prédéfinies, leur définition et les techniques pour les utiliser. Sauf à souhaiter rester dans la catégorie des débutants, il faut mémoriser rapidement leur syntaxe. On utilise l’aide en cas d’oubli.

Excel offre des moyens variés pour l’utilisation des fonctions prédéfinies : pour trouver les fonctions il faut aller dans l’onglet Accueil dans le groupe Edition : on a alors le symbole ∑ qui donne directement la fonction Somme en cliquant dessus.

 (
En cliquant sur le symbole 
S
omme, il écrit directement dans la cellule active 
 
=somme(
G2
). Il a déterminé lui-même les cellules adjacentes
,
 ici 
G2
 qu’on peut modifier en sélectionnant immédiatement une autre plage
. Il indique la syntaxe en-dessous
Si on positionne le curseur sur le nom de la fonction en bleu
,
 il prend la forme d’une main et un cli
c
 ouvr
e l’aide concernant la fonction
)[image: ]

En cliquant sur la flèche vers le bas à côté du signe ∑, il déroule une liste avec les fonctions les plus utilisées et se terminant par Autres fonctions… qui donne accès à un assistant permettant de trouver les différentes fonctions regroupées en catégories.
 (
On a cliqu
é sur la flèche et choisi
 dans le menu déroulant Autres fonctions
 :
L’assistant « Insère
r
 une fonction » s’ouvre 
Dans la cellule active le signe 
 
= est écrit
) (
Dans l’assistant « Inserer » on peut taper une description de la fonction recherchée, ou sélectionner une catégorie par 
la liste déroulante
Choisir une fonction d
ans la liste. Sa syntaxe apparaît en-
dessous.
On peut obtenir une aide plus complète sur la fonction
)[image: ]

Note
Si on a choisi une fonction pour une cellule active et que la saisie n’a pas été validée en choississant Autres fonctions…, on obtient l’aide de la fonction choisie.

Vous avez aussi l’accès au fonctions par l’onglet  Formules dans le premier groupe Bibliothèques de fonctions, bouton Insérer une fonction.
 
 (
Bascule permettant d’afficher les formules ou les valeurs dans les cellules du dossier
) (
Les fonctions sont directement regroupées en domaines
 : 
financières, logiques, texte,…
)[image: ] 

Compte tenu de l’utilisation des fonctions dans les formules, Excel a implanté sous forme de fonctions des éléments de programmation classique, notamment pour les opérateurs logiques et le branchement conditionnel.

En algorithmique, on introduit le « si » immédiat de syntaxe suivant :
si(<condition>, <valeur1>, <valeur2>)
C’est une fonction à trois paramètres dont le premier est une condition et qui retourne la valeur 1 si la condition est vraie et sinon la valeur 2.

Dans Excel elle est implantée avec la syntaxe suivante :
SI(<condition>;<valeur1>;<valeur2>)

Exemple	Si A9 contient la moyenne d’un étudiant en B9, on pourra mettre la formule :
=SI(A9>=10 ;"ADMIS";"ECHEC")
pour les textes « en dur » (et dates) les valeurs doivent être entre guillemets.

Note
Dans les langages de programmation ou les logiciels ayant des fonctions prédéfinies, les paramètres d’une fonction sont en principe séparés par des virgules. Dans la version française d’Excel, les virgules servant de séparateur décimal, on utilise le point virgule.

On dispose des opérateurs relationnels suivants :

	Opérateurs
	Notations

	Inférieur ou égal
	<=

	Inférieur
	<

	Supérieur ou égal
	>=

	Supérieur
	>

	Différent
	<>


Exemple
 (
Formule correspondant à la colonne B
=SI(A3>=16;"TB";SI(A3>=14;"B";SI(A3>=12;"AB";SI(A3>=10;"P";"Ajourné"))))
) (
La formule 
pour la colonne C 
est la suivante :
=SI(A3>=10;"ADMIS";"ECHEC")
puis on a recopié en-dessous
)[image: ]

La formule dans la barre de saisie est plus complexe. La valeur 2 est remplacée par le résultat d’un autre SI. Ainsi si A3 est supérieur à 16, c’est TB ; sinon on regarde si c’est supérieur ou égal à 14 et ainsi de suite.

Rappel
Dans le calcul propositionnel, on considère qu’une assertion P peut être soit vraie, soit fausse mais pas les deux à la fois. A partir de deux assertions notées P et Q on peut construire :
l’assertion « P et Q » qui est vraie si les deux sont vraies et fausse dans tous les autres cas,
l’assertion  «  P ou Q » qui est fausse si les deux sont fausses et vraie dans les autres cas (au moins une est vraie),
l’assertion « Non(P) » qui est vraie si P est fausse et fausse si P est vraie.

Enfin on a les lois de Morgan suivantes :
Non(P et Q)  Non(P) ou Non(Q)
Non(P ou Q)  Non(P) et Non(Q)
ET, OU, NON sont dits des opérateurs logiques.

Ce calcul booléen nous intéresse car dans le SI la condition est une assertion. Elle peut être construite à partir des opérateurs logiques. Sous Excel les opérateurs logiques sont implantés sous forme de fonctions.

La fonction ET accepte un nombre quelconque (maximum 255) d’expressions logiques. Elle retourne la valeur VRAI si chaque argument est vrai et FAUX sinon.
Sa syntaxe est :
ET(<expr logique1>;<expr logique2>;< expr logique3>;…)

Exemple (dans une cellule autre que A1)
=ET(A1>=0 ;A1<=20)  renvoie VRAI si A1 est compris au sens large entre 0 et 20
=SI ( ET(A1>=-1; A1<=1) ; "plus petit que 1 en valeur absolue" ; "plus grand que 1 en valeur absolue")

La fonction OU accepte un nombre quelconque (maximum 255) d’expressions logiques ; elle retourne la valeur vraie si au moins un argument est vrai et fausse sinon (ie tous les arguments sont faux).
Sa syntaxe est :
OU(<expr logique1>;<expr logique2>;< expr logique3>;…)

La fonction NON accepte un argument de type logique est retourne VRAI si l’argument est faux et faux sinon.
Sa syntaxe est :
NON(<expression logique>)
(il existe deux autres fonctions FAUX() qui donnent toujours faux et VRAI() qui donnent toujours vrai)

Exemple
On trouvera ci-dessous les tables de vérité extraites d’Excel concernant le calcul propositionnel avec l’affichage des formules pour l’un (onglet Formules groupe Audit de formules) et l’autre donnant les valeurs :

	
	F
	G
	H
	I
	J
	K

	1
	P
	Q
	P et Q
	P ou Q
	Non (P et Q)
	Non (P) ou non(Q)

	2
	=VRAI()
	=VRAI()
	=ET(F2;G2)
	=OU(F2;G2)
	=NON(H2)
	=OU(NON(F2);NON(G2))

	3
	=VRAI()
	=FAUX()
	=ET(F3;G3)
	=OU(F3;G3)
	=NON(H3)
	=OU(NON(F3);NON(G3))

	4
	=FAUX()
	=VRAI()
	=ET(F4;G4)
	=OU(F4;G4)
	=NON(H4)
	=OU(NON(F4);NON(G4))

	5
	=FAUX()
	=FAUX()
	=ET(F5;G5)
	=OU(F5;G5)
	=NON(H5)
	=OU(NON(F5);NON(G5))



	
	F
	G
	H
	I
	J
	K

	1
	P
	Q
	P et Q
	P ou Q
	Non (P et Q)
	Non (P) ou non(Q)

	2
	VRAI
	VRAI
	VRAI
	VRAI
	FAUX
	FAUX

	3
	VRAI
	FAUX
	FAUX
	VRAI
	VRAI
	VRAI

	4
	FAUX
	VRAI
	FAUX
	VRAI
	VRAI
	VRAI

	5
	FAUX
	FAUX
	FAUX
	FAUX
	VRAI
	VRAI

On trouve en particulier :

Les fonctions trigonométriques classiques dont le paramètre est une valeur ou une référence. On a par exemple COS pour cosinus, COSH pour cosinus hyperbolique, ACOS pour arc cosinus, ACOSH pour arc cosinus hyperbolique et les autres avec des notation semblables (A pour arc, H pour hyperbolique et les trois premières lettres de la fonction). On peut ajouter la fonction DEGRES qui convertit des radians en degrés.
Les fonctions mathématiques classiques ABS pour valeur absolue, EXP pour la fonction exponentielle, LN pour le logarithme népérien, LOG (valeur;base) pour le logarithme dans une base et LOG10 pour le logarithme en base 10.
Les fonctions PUISSANCE (nombre;puissance), RACINE (nombre) pour la racine carrée, PRODUIT (nombre1;nombre2;...) pour le produit des nombres donnés en argument (pour la fonction ALEA voir paragraphe sur les fonctions statistiques).
Les fonctions sur les arrondis : elles permettent d’arrondir la valeur mémorisée et sont indépendantes du format d’affichage qui n’intervient que sur la valeur affichée dans la cellule, « indépendant » ne signifiant pas « incohérent » ! Il existe plusieurs fonctions pour arrondir compte tenu : - d’une part que les nombres sont positifs ou négatifs - et d’autre part qu’on peut vouloir arrondir au plus proche par valeur inférieure, supérieure, vers zéro, en s’éloignant de zéro ou à un multiple près.
Exemple
On veut un coupon de tissu de 12,3 m et on ne peut acheter que des coupons de longueur multiple de 10 mètres, ici donc 20 mètres ; ou une longueur de 15 mètres de tuyau vendu par morceaux de 10 mètres. Remarquons que « par valeur supérieure » entraîne sur 8,5 un arrondi à 9 et sur -8,5 un arrondi à -8, «  en s’éloignant de zéro » (concept que l’on ne trouve pas en mathématiques) entraîne un arrondi de 9 pour 8,5 et -9 pour -8,5. Ces fonctions ont généralement deux arguments ou paramètres : la valeur qu’on veut arrondir et la précision de l’arrondi indiquée par le nombre de chiffres après la virgule. La liste de ces fonctions est : ARRONDI, ARRONDI.AU.MULTIPLE, ARRONDI.INF, ARRONDI.SUP, ENT (partie entière), IMPAIRE, PAIRE, PLAFOND, PLANCHER, TRONQUE.
On trouve la fonction très utile SOMME(<une sélection>) qui calcule la somme des cellules de type numérique de la sélection. Le bouton ∑ sur une cellule détermine la sélection la plus vraisemblable dans la feuille, c’est-à-dire la somme de la colonne au-dessus ou de la ligne si la colonne est vide.
Attention !
On peut aussi utiliser cette fonction sans référence de sélection, mais avec des valeurs . Dans ce cas il y a une conversion des valeurs logiques en numérique qui vaut 1.
On a aussi la fonction SOMME.SI qui fait intervenir une plage à sommer B1:B10. Les valeurs de chaque ligne seront additionnées si dans la ligne correspondante d’une autre plage la valeur vérifie un critère.
La syntaxe est :
=SOMME.SI(<autre plage>;«critère»;plage à sommer)



Exemple
Pour calculer la somme des valeurs de la plage B1:B10 si dans la même ligne en colonne A la valeur est supérieure à 100, on utilisera la formule :
=SOMME.SI(A1:A10;">100";B1:B10)

[bookmark: _Toc94001846][bookmark: _Toc251074965][bookmark: _Toc251076470][bookmark: _Toc251076863][bookmark: _Toc251576518]Fonctions Statistiques
Elles sont nombreuses et comprennent un grand nombre de lois de probabilités. Le lecteur doit connaître (ie taper directement au clavier) les fonctions donnant la moyenne (MOYENNE), la variance (VAR.P), la variance corrigée (VAR), l’écart type (ECARTYPEP) et celui corrigé (ECARTYPE), la médiane (MEDIANE), le nombre de données (NB) et toutes celles correspondant à ses connaissances en statistiques.

La fonction ALEA apparaît sous Excel dans la catégorie Mathématiques. Elle fournit un nombre au hasard entre 0 et 1 (conformément à la loi uniforme sur cet intervalle). Cette fonction est importante car les probabilistes ont montré que si on sait simuler une loi uniforme sur 0,1 alors on peut simuler toute autre loi. Elle est donc très utile dans les logiciels de Jeux. La validation précédée de l’emploi de F9 ne copie pas une formule dans la cellule mais la valeur, ainsi la formule disparaît.

Signalons la fonction NB, la fonction NB.SI (plage;critère), et NB.VIDE comptant les cellules vides.

[bookmark: _Toc94001847][bookmark: _Toc251074966][bookmark: _Toc251076471][bookmark: _Toc251076864][bookmark: _Toc251576519]Fonctions Texte
Les fonctions texte comprennent par exemple la fonction CONCATENER, mais aussi les fonctions cherchant la position d’une sous-chaîne de caractères dans une chaîne de caractères, le remplacement d’une partie d’un texte par un autre. La conversion en texte d’une valeur numérique et réciproquement, la suppression des espaces, etc.

Le calcul matriciel est implanté sous forme de fonction dans cette catégorie. Ces fonctions calculent ou manipulent des tableaux de chiffres aux dimensions fixées qui doivent être traités dans leur ensemble et sont insécables. Leur utilisation ne peut donc se faire en général dans une seule cellule. Il faut sélectionner la plage des résultats puis taper la formule dans la première cellule et la valider avec Ctrl + Maj+ Entrée. On trouve en particulier dans les fonctions statistiques l’utilisation des matrices notamment dans la fonction FREQUENCE qui exige ce type de saisie et de validation.

Exemple (tiré d’un exercice de deuxième année du DUGEAD)

En colonne A on a des données comprises entre 0 et 19. En colonne C on a les bornes des classes correspondantes aux intervalles ]-0,01..4],]4..5.5], .., ]16..19]. En colonne D on veut calculer les effectifs de chaque classe.
 (
On sélectionne la plage où seront les effectifs
On tape la formule
Il reste à la valider par 
Ctrl
+
Maj
+
Entrée
 
On doit obtenir la colonne F
)[image: ]

Autre technique : on a sélectionné uniquement  D et on a tapé la formule en la validant classiquement. A l’aide du clavier on effectue la sélection de la plage D8:D17. On tape sur la touche F1 qui nous passe en mode modification et on valide par Ctrl +Maj+Entrée.
[bookmark: _Toc94001849][bookmark: _Toc251074968][bookmark: _Toc251076473][bookmark: _Toc251076754][bookmark: _Toc251076866][bookmark: _Toc251576521]Autres catégories de fonctions
Les financiers trouveront les fonctions classiques qu’ils utilisent.
Enfin Excel offre la possibilité de traiter les bases de données, bien qu’un logiciel de base de données soit préférable.

[bookmark: _Toc94001851][bookmark: _Toc251074970][bookmark: _Toc251076475][bookmark: _Toc251076756][bookmark: _Toc251076868][bookmark: _Toc251576523]Listes algorithmiques et Recherche séquentielle
Une liste est une suite finie de valeurs. Quand les valeurs sont de même type, on dit qu’elle est homogène et sinon qu’elle est hétérogène. La manipulation des listes repose sur 3 concepts :
la longueur de la liste, c’est-à-dire le nombre de places dans la liste, le numéro des places et la valeur à chaque place.

Exemple
L :=["André", "David", "Florian", "Julien", "Mathieu", "Oriane", "Sébastien", "Sophie"]
C’est une liste homogène à 7 éléments. Le premier est « André » ;
le troisième a pour valeur « Florian » ; « Mathieu » est le cinquième élément de la liste.

Toute plage sur une même ligne (exemple A1:C1) ou sur une même colonne (exemple A3:A8) est de fait une liste. Une plage quelconque (exemple A1:C3) est considérée comme une liste dès qu’on a choisi un parcours par ligne.

Dans notre exemple, elle correspond à la liste des valeurs des références suivantes : (A1,B1,C1,A2,B2,C2,A3,B3,C3) ; ou par colonne elle correspond à la liste des valeurs des références suivantes : (A1,A2,A3,B1,B2,B3,C1,C2,C3).



Note
Sous Excel le mot liste a une signification différente bien qu’un informaticien considère que le terme est correct.

Pour une liste donnée, on dit qu’on effectue une recherche séquentielle d’une valeur X dans une liste si partant d’un élément de la liste, généralement le premier, on recherche la première valeur de la liste correspondant à la valeur cherchée X en comparant les éléments de la liste les uns après les autres. Le processus s’arrête dès qu’une occurrence est trouvée dans la liste. La valeur recherchée peut apparaître plusieurs fois. On peut chercher l’occurrence suivante.

Il existe deux options :
une recherche exacte (la valeur cherchée est égale à une valeur ou contenue dans une valeur),
une recherche approchée (correspond au début d’une valeur ou à la plus grande valeur inférieure à la valeur cherchée).

Remarque
Une recherche tient compte ou non de la casse (majuscule, minuscule).
Une recherche s’effectue en boucle, c’est-à-dire reprend au début de la liste.


[bookmark: _Toc94001854][bookmark: _Toc251074973][bookmark: _Toc251076478][bookmark: _Toc251076871][bookmark: _Toc251576526]
Recherche séquentielle sous Excel

Principe
Si on n’a aucune sélection, la liste est constituée de toutes les cellules non vierges de la feuille rangées par lignes les unes après les autres ou selon l’option choisie par colonnes les unes après les autres. La recherche sera exacte si on coche « cellule entière » ; dans ce cas même un espace à la fin d’une valeur est pris en compte et peut faire échouer la recherche. On dispose aussi de l’option « Respecter la casse ».
	Opérationnellement on peut utiliser dans l’onglet Accueil/groupe Edition, le raccourci Ctrl + F ou encore Maj +F5. La boîte de dialogue suivante s’ouvre :

 (
On choisit le parcours par ligne ou colonne
) (
Case à cocher :la recherche se fait sur la valeur exacte
) (
On tape la valeur recherchée
)[image: ]

Le bouton « suivant » cherche l’occurrence suivante. Mais il est plus simple de fermer, puis de taper le raccourci Maj+F4. Pour faire une recherche en arrière, il suffit de taper Ctrl+Maj+F4. « Dans » permet de limiter la recherche uniquement aux valeurs ou aux formules ou encore aux commentaires, mais cette option réserve quelques surprises. L’option « valeur » comprend aussi les valeurs affichées par les formules et l’option « Formule » recherche dans les données et les formules, mais pas dans la valeur affichée d’une formule.
Remarque
Si on a fait une sélection, la recherche est limitée à la sélection.
[bookmark: _Toc94001855][bookmark: _Toc251074974][bookmark: _Toc251076479][bookmark: _Toc251076758][bookmark: _Toc251076872][bookmark: _Toc251576527]FONCTION DE RECHERCHE RAPIDE

Pour la recherche rapide d’un élément dans une liste, on suppose que la liste est triée par valeur croissante. En effet sur les listes triées, on utilise l’algorithme dichotomique suivant pour chercher une valeur A :
on prend la valeur à la moitié de la liste qu’on compare à la valeur cherchée A. Soit elle est avant soit après. Il suffit donc de recommencer sur la moitié correspondante et ainsi la liste diminue de moitié à chaque fois. On remarque que rechercher une valeur sur une liste de 20 000 éléments nécessite une comparaison supplémentaire par rapport à une liste de 10 000 éléments.

[bookmark: _Toc94001858][bookmark: _Toc251074977][bookmark: _Toc251076482][bookmark: _Toc251076875][bookmark: _Toc251576530]Principe
Excel accepte de trier des listes hétérogènes de valeurs avec la convention suivante : les valeurs numériques par ordre croissant sont triées en premier puis les chaînes de caractères par ordre alphabétique (alphanumérique), puis les valeurs booléennes, le FAUX étant avant le VRAI et enfin les cellules vides. Pour les valeurs alphabétiques on peut ou non tenir compte des minuscules ou des majuscules, on dit « respecter la casse ».
	Si on a des valeurs égales, la première de la liste à trier apparaîtra en premier, la deuxième en deuxième etc. ; mais Excel permet de préciser un deuxième et même un troisième critère en cas d’homonymes. Par exemple si on a les noms des étudiants en colonne A et les prénoms en colonnes B, on peut demander un tri sur la colonne A et en cas d’homonyme utiliser la colonne B des prénoms. En informatique le premier critère est dit clé primaire, le second pour les homonymes clé secondaire. 

	Si on sélectionne une seule colonne, et qu’on demande un tri, compte tenu que les valeurs de cette colonne vont changer de ligne lors du tri, Excel examine si les colonnes adjacentes à la sélection sont remplies et propose de les changer aussi de ligne ou de colonne. C’est l’option « Etendre la sélection ». La sélection est étendue jusqu'à une colonne et une ligne entièrement vide bordant la sélection. Le message est le suivant :

[image: ]
Si on veut trier uniquement la selection il faut choisir « Continuer avec la sélection en cours », dans ce cas l’ensemble des cellules non sélectionnées sera inchangé.
	Enfin pour le tri sur une colonne, Excel considère par défaut que le premier élément de la colonne peut être un titre de colonne et qu’il ne doit pas être modifié par le tri. Il suffit de choisir de cocher mes données ont des en-têtes. On a la même démarche pour un tri sur une ligne.

Dans l’exemple ci-dessous, on a recopié la colonne A et B en C :D et aussi en E :F. On a sélectionné C2 à D10. La sélection comprenant 2 colonnes, le message d’extension de la sélection ci-dessus n’est pas proposé, puis on a choisi l’onglet Accueil/groupe Edition et Tri personnalisé sans l’option Respecter la casse. Les colonnes F et G avec les mêmes données ont été triées avec l’option Respecter la casse.

 (
On retrouve les nombres en premier, les chaînes de caractères sans tenir compte des minuscules. Tous les « Andre » sont classés en fonction des prénoms deuxième clé 
) (
On a fait le même tri mais avec l’option « respecter la casse » les minuscules étant avant les majuscules « ANDre » est devant les autres « ANDRE »
) (
On choisit T
ri personnalisé permettant le choix des options pour les deux tris. Les colonnes A et B sont inchangées
)[image: ]

Le tri personnalisé donne accès à la fenêtre de dialogue suivante :


 (
En cliquant sur option, la fenêtre Options du tri apparaît.
Ici on a décoché
 R
especter la casse
et on a choisi
 par colonne
) (
On a coché
 Mes données ont des en-têtes ; l
a sélection ne comporte plus le titre L
iste recopiée et triée
On a ajouté une clé par un clic sur Ajouter un niveau et on a fixé la colonne D
Les deux flèches inversent les clés
On a chois
i
 le tri sur les valeurs, mais on peut aussi sur la couleur de la cellule, de
 la police ou par icône
)[image: ]


Excel détermine les titres de la façon suivante :
si la ligne au-dessus n’est pas vide il prend cette ligne comme ligne de titres,
sinon il prend la première ligne de la plage. 

[bookmark: _Toc251074979][bookmark: _Toc251076484][bookmark: _Toc251076877][bookmark: _Toc251576532]Choix : trier de A à Z


Sur une sélection, le tri s’effectue sur la première colonne uniquement. Si les valeurs de chaque colonne de la sélection sont du même type, alors le tri s’effectue avec l’option Ligne de titres à « Non », sinon il considère que la première ligne est sans doute une ligne de titres, et ne prend pas en compte la casse.

Note
Si le tri ne correspond pas à celui recherché, on peut toujours l’annuler par Ctrl Z (menu Edition) et recommencer en apportant les modifications nécessaires sur la plage et les choix du tri avec le tri personnalisé.

Syntaxe
CHOISIR(numéro de l’emplacement ; valeur1 ; valeur2 ; valeur3 ; valeur4)
Donne la valeur correspondant au numéro de la place.
Le numéro peut être une référence ; « valeur » peut être remplacée par des références.

Exemple
	=CHOISIR(2;"HOMME";"FEMME")
affiche FEMME dans la cellule car c’est le deuxième élément

	=CHOISIR(C5;"HOMME";"FEMME")
affiche HOMME si la cellule C5 contient 1 ou FEMME si C5 contient 2

	=CHOISIR(3 ;A1 ;A2 ;A3)
affiche la valeur de la cellule A3 car la troisième valeur étant A3 il évalue alors =A3

	=SOMME(CHOISIR(2,A1:A5 ; B1:B5 ; C1:C5)
donne B1:B5, il évalue alors SOMME(B1:B5)
[bookmark: _Toc94001863]

Principe
Elle fonctionne avec 3 paramètres : la valeur X à chercher, une première liste triée et une deuxième liste de même longueur. Elle cherche la valeur X dans la première liste triée, une fois trouvée, elle mémorise le numéro de la place correspondante et elle donne la valeur de la deuxième liste ayant la même place. Ainsi si on cherche "Florian" dans la liste ["André", "David", "Florian", "Julien", "Mathieu", "Oriane", "Sébastien", "Sophie"] et que la troisième est [12,18,13,14,10,8,12,10] : Excel trouve la valeur à la troisième place dans la première liste et donne donc 13, troisième valeur de la deuxième liste. 

	S’il ne trouve pas la valeur exacte, il donne la plus grande inférieure. Si on cherche "Henry" la plus grande valeur inférieure est toujours "Florian" et donc on obtient toujours 13. La valeur cherchée doit donc être supérieure ou égale au premier 

Syntaxe
RECHERCHE(valeur_cherchée;vecteur_recherche;vecteur_résultat)
Chaque argument peut être des références, le vecteur recherche et le vecteur résultat sont de même dimension, uni ligne ou uni colonne indépendamment l’un de l’autre.

Note
Cette fonction peut aussi travailler avec des matrices. Il existe aussi deux autres fonctions RECHERCHEH et RECHERCHEV dont le principe général est le même, mais travaillant respectivement sur des colonnes ou des lignes ; on peut fixer une option exacte ou approchée (voir l’aide en ligne).

